NASA Technical Memorandum 85651

Development of a Nuclear Technique for Monitoring Water Levels in Pressurized Vessels

Jag J. Singh, William T. Davis, and Gerald H. Mall

SEPTEMBER 1983

NASA

NASA Technical Memorandum 85651

Development of a Nuclear Technique for Monitoring Water Levels in Pressurized Vessels

Jag J. Singh and William T. Davis
Langley Research Center
Hampton, Virginia

Gerald H. Mall
Computer Sciences Corporation
Hampton, Virginia

NASA
National Aeronautics
and Space Administration

Scientific and Technical
Information Branch

1983

SUMMARY

A new technique for monitoring water levels in pressurized stainless steel cylinders has been developed. It is based on the differences in gamma ray attenuation coefficients in water and air. For an open-face, full-scale-model steel cylinder and a 10-μCi Cs^{137} gamma ray source, the counting rates for air and water differ by a factor of 15.60 ± 0.60 for cutoff energy of 200 keV and by a factor of 17.24 ± 1.40 for cutoff energy of 511 keV. These results are in reasonable agreement with the respective calculated values of 15.98 ± 1.50 and 17.68 ± 1.95. The respective values when the air is pressurized to 408.2 atm are calculated to be 4.75 ± 0.51 and 5.01 ± 0.64. This large difference in counting rates ensures a clear capability for detecting the presence or absence of water in the gamma ray path in pressurized vessels. Computer programs for calculating nuclear radiation transmission are included in the appendix.

INTRODUCTION

Thermal probes in the Langley 8-Foot High-Temperature Tunnel have to be water-cooled in order to protect them from burnup during tunnel runs. The cooling water is stored in 30-ft-high, 16-in-diameter steel cylinders. Figure 1 shows a general view of the cylinder assembly for storage of cooling water. The water level in the cylinder is initially monitored using a simple U-tube manometer. Before the start of a tunnel run, high-pressure (about 408 atm) air is introduced into the cylinder above the water level and forces the cooling water through the multijacketed thermal probes. Currently, there are no means available for directly monitoring the water level in the reservoir cylinders once the cylinders are pressurized. It is presumed that the initial manometer reading in the "nonpressurized" state gives correct indication of the initial water level in the reservoir cylinder. This, of course, is true only if there was no residual pressurized air above the water level during the water pressure reading. It is therefore necessary to develop an independent technique that monitors the presence of water at a critical height in the pressurized water reservoir cylinder and thus ensures an adequate supply of cooling water for the duration of the test. One technique that promises to meet this objective is based on differences in gamma ray attenuation coefficients in water and air. The principle of this technique as well as experimental results obtained with a full-scale mock-up model are described in the following sections.

SYMBOLS

c	velocity of light
D	internal diameter of the steel reservoir cylinder
E_{dis}	cutoff energy
h	Planck's constant
I_t	transmitted intensity

I_o incident intensity

m_o rest mass of the electron

r_o classical radius of the electron

w_i fraction by weight of the ith element

x path length through a medium

Z atomic number

θ scattering angle

μ total linear attenuation coefficient of the medium for the incident photons

μ_c attenuation coefficient for the Compton process, given by the product of atomic density and Compton scattering cross section per atom

μ_i linear attenuation coefficient for the ith element

μ_{pe} attenuation coefficient for the photoelectric effect, given by the product of atomic density and the photoelectric cross section per atom

μ_{pp} attenuation coefficient for the pair production process, given by the product of atomic density and pair production cross section per atom

ν_o incident photon frequency

ν' scattered photon frequency

ρ density

ρ_i density of the ith element

Ω solid angle

Subscripts:

a air

s steel

w water

PRINCIPLE OF OPERATION

When a collimated beam of photons traverses a medium, it can interact with the atoms of the medium in a number of ways, depending upon its energy. These interactions result in the removal of the photon from the incident beam. The transmitted

intensity I_t of a collimated beam of photons of incident intensity I_o after traversing through a thin absorber of thickness x may be written as follows (ref. 1):

$$I_t = I_o e^{-\mu x} \tag{1}$$

where μ is the total linear attenuation coefficient of the medium for the incident photons. μ includes the effects of all types of interactions that the incident photons suffer in the absorber. For photons in the energy range 0.1 to 10.0 MeV, the dominant mechanisms are the photoelectric effect, the Compton effect, and the pair production effect; i.e.,

$$\mu = \mu_{pe} + \mu_c + \mu_{pp} \tag{2}$$

where

μ_{pe} = Attenuation coefficient for the photoelectric effect, given by the product of atomic density and photoelectric cross section per atom

μ_c = Attenuation coefficient for the Compton process, given by the product of atomic density and Compton scattering cross section per atom

μ_{pp} = Attenuation coefficient for the pair production process, given by the product of atomic density and pair production cross section per atom

Figure 2 shows the relative importance of these effects for various elements and for various photon energies (ref. 1). It is apparent that for $Z \leqslant 40$, the dominant mechanism for gamma ray energies of 0.2 to 7.0 MeV is the Compton scattering effect.

The values of μ for different absorbers for photons in the energy range 0.01 to 100.00 MeV have been reported by several authors (refs. 2 and 3). If the absorber is made up of a compound or a mixture of several elements, the overall attenuation coefficient is given by the following relation:

$$\frac{\mu}{\rho} = \frac{\mu_1}{\rho_1} w_1 + \frac{\mu_2}{\rho_2} w_2 + \frac{\mu_3}{\rho_3} w_3 + \cdots \tag{3}$$

where

ρ_i = Density of the ith element

μ_i = Linear attenuation coefficient for the ith element

w_i = Fraction by weight of the ith element

If the absorber, on the other hand, is made up of a series of physically separated materials, the transmission process through each material should be treated separately. Equation (1) in that case will take the following form:

$$I_t = I_o e^{-\sum_i \mu_i x_i} \tag{4}$$

where

μ_i = Linear attenuation coefficient for the gamma rays in the ith component of distributed absorber

x_i = Path length in the ith component of the distributed absorber

In the specialized case of a pressurized water vessel, equation (4) can be rewritten as follows:

$$I_t(\text{water}) = I_o e^{-(2\mu_s x_s + \mu_w x_w)} \tag{5}$$

$$I_t(\text{air}) = I_o e^{-(2\mu_s x_s + \mu_a x_a)} \tag{6}$$

where

μ_s = Linear attenuation coefficient for the gamma rays in steel

μ_w = Linear attenuation coefficient for the gamma rays in water

μ_a = Linear attenuation coefficient for the gamma rays in air

$2x_s$ = Gamma ray path length in steel (twice the steel cylinder wall thickness)

x_w = Gamma ray path length in water (internal diameter of steel cylinder)

x_a = Gamma ray path length in air (internal diameter of steel cylinder)

Combining equations (5) and (6), one obtains

$$\frac{I_t(\text{air})}{I_t(\text{water})} = e^{-(\mu_a x_a - \mu_w x_w)}$$

$$\frac{I_t(\text{air})}{I_t(\text{water})} = e^{D(\mu_w - \mu_a)} \tag{7}$$

where D is the internal diameter of the steel reservoir cylinder. The proposed technique is based on the fact that μ_w and μ_a are very different for photons in the energy range 0.2 to 7.0 MeV. As indicated earlier, the linear attenuation coefficient of a medium is a very strong function of the photon energy. In order to obtain a reasonably high value of $I_t(\text{air})/I_t(\text{water})$ consistent with the reliability of operation of the monitor, Cs^{137} was selected as the gamma ray source. Cs^{137} has a half-life of 30 years and emits a single gamma ray of energy 662 keV. As seen from figure 2, the dominant mode of interaction at this energy in all components of the pressurized stainless steel water vessel would be Compton effect.

A theoretical discussion for the specific case of transmission of Cs^{137} (662 keV) gamma rays through the pressurized, stainless steel, cooling water cylinder is given below. Figure 3 shows the essential geometry of the problem.

Figure 4 illustrates the Compton scattering process for the case when the incident photon energy far exceeds the binding energy of the struck electron. The differential Compton collision cross section per electron for incident unpolarized radiation of energy $h\nu_o$ is given by the following equation (refs. 1, 4, and 5)

$$\frac{d\sigma_c}{d\Omega} = \frac{r_o^2}{2}\left(\frac{\nu'}{\nu_o}\right)^2\left(\frac{\nu_o}{\nu'} + \frac{\nu'}{\nu_o} - \sin^2\theta\right) \qquad \text{cm}^2/\text{electron} \tag{8}$$

where the scattered photon $h\nu'$ goes into the solid angle $d\Omega = 2\pi \sin\theta\, d\theta$, and r_o is the classical electron radius (2.818×10^{-13} cm).

Because of the finite degree of collimation of the incident beam, finite thicknesses of various components of the absorber medium in the path of the beam, and finite size of the radiation detector, multiple scattering effects must be taken into account (ref. 6). Because of the complexity of the secondary radiation produced in each collision, accurate theoretical calculation of broad beam attenuation is very difficult. Several different procedures have been used to calculate gamma ray transport through thick absorbers. Among the prominent methods reported (ref. 6) are (1) the method of successive scatterings, (2) the method of moments, and (3) the method of random sampling (Monte Carlo). A Monte Carlo approach (refs. 6 and 7) was utilized in the present study in order to determine the number of photons, starting from a collimated Cs^{137} source, that arrive at the uncollimated detector with an Energy $\geq E_{dis}$. The theoretical details of the computational procedure are summarized in the appendix. The inclusion of multiple scattering effects reduces the difference in the counting rates between air and water media in the pressure vessel. For example, for air at a pressure of 1 atm, the ratio $I_t(\text{air})/I_t(\text{water})$ falls from 20.17 to 17.68 ± 1.95 when only scattered photons of Energy $\geq$ 511 keV are counted and further falls to 15.98 ± 1.50 when the scattered photon energy discriminator level is reduced to 200 keV. For air at 408.2 atm, the corresponding values are 5.26, 5.01 ± 0.64, and 4.75 ± 0.51, respectively.

EXPERIMENTAL RESULTS

A full-scale laboratory prototype system was constructed in order to test the concept of a nuclear technique for monitoring water levels. Figure 5 shows a schematic diagram of the experimental system for monitoring the presence or absence of

water medium in the path of the test gamma rays. A 10-microcurie (nominal) Cs^{137} source provided the test gamma rays. The gamma rays transmitted through the pressure vessel were counted with a 2-in-diameter, 2-in-high NaI(Tl) crystal mounted on a high-gain photomultiplier. The photomultiplier output, after suitable amplification, was passed through a single-channel analyzer. Counts were recorded for predetermined intervals for two discriminator settings with and without water in the steel cylinder. The first discriminator setting of 200 keV was selected to provide for counting of all photons, having energies above the photoelectric energy limit, that arrived at the radiation detector. The counts registered will reflect considerable impact of multiple scattering effects in the absorbers between the source and the detector. The second discriminator setting of 511 keV was selected to restrict the counts to those photons that had suffered minimal scattering and, consequently, had lost minimal energy. The average values of $I_t(air)/I_t(water)$ for the two discriminator settings are summarized in table I. These values are in reasonable agreement with the corresponding calculated values listed in table II.

Main reasons for the rather large errors in the experimental values lie in the weakness of the Cs^{137} test source. For example, for E_{dis} = 511 keV, the counting rates with an empty cylinder with and without the source were approximately 740 and 418 counts per minute (cpm), respectively, whereas the counting rate with the source and with water in the cylinder was only 437 cpm. This poor signal-to-background ratio with water in the cylinder produces a large error in the $I_t(air)/I_t(water)$ value. A stronger test source should provide much better counting statistics and hence smaller error. This conclusion was verified by constructing a half-scale model and repeating all the aforementioned tests with it. The comparison between the experimental results and the calculated values is summarized in table III. The errors in the experimental results this time are considerably lower, as expected.

Despite the large error in the $I_t(air)/I_t(water)$ value with E_{dis} = 511 keV in the full-scale model results, this discriminator setting would be preferable with a stronger radiation source because it makes the monitor more sensitive to the presence or absence of water in the test radiation beam path.

The prototype system, of course, could not be pressurized to 408 atm for reasons of safety. It was therefore not possible to determine experimentally the value of $I_t(air\text{-}pressurized)/I_t(water)$. However, there is no reason to doubt the theoretically calculated value, since the theory of gamma ray transmission through absorbers of various thicknesses is well understood.

Figure 6 shows the experimentally observed counting rate for E_{dis} = 511 keV as the water was allowed to escape from the test cylinder. It is noted that the counting rate rises from an average value of 437 cpm to 740 cpm as the water is replaced by the air in the path of the test gamma rays. Allowing for a background counting rate of 418 cpm, these measurements give a value of approximately 17 for $I_t(air)/I_t(water)$ in reasonable agreement with the average value of 17.24 ± 1.40 obtained under static conditions.

In order to ensure that the tunnel could not be operated without adequate water supply in the pressurized water cylinders, a safety circuit based on the Cs^{137} counting rate change has been designed, and a breadboard prototype has been built and successfully tested. The electronic safety circuit diagram is shown in figure 7. The circuit, besides providing a visible flashing light signal, will also provide an audio alarm signal if the water level is inadequate to start the tunnel operation. The pulses from the single-channel analyzer are shaped by the one-shot logic circuit

which increases the pulse width from 1 μs to 1.3 ms. This pulse width will permit handling counting rates as high as 769 counts per second (cps). Higher counting rates will require shorter pulse width. The integration of these pulses produces a dc voltage proportional to the counting rate. The comparator circuit will turn on the alarm as this dc voltage matches or exceeds a preset reference level.

CONCLUDING REMARKS

The full-scale prototype system results under a pressure of 1 atm clearly demonstrate that the nuclear technique for monitoring water levels should easily distinguish between water and air paths inside the water reservoir cylinders. Since the pressurized water cylinder tests could not be conducted with a laboratory prototype system, full-scale tests on actual cooling water cylinders were not performed. However, reasonably good agreement between theoretical calculations and the experimental results for an open-faced prototype system indicates that equally good agreement should be obtained for a pressurized system, since the theory of electromagnetic radiation interaction with matter is well understood.

The technique should, of course, be usable with other fluids, since the linear attenuation coefficients for intermediate energy gamma rays in air are considerably lower than in the fluids. It should also be adaptable to monitor continuously the fluid level in the reservoir systems and underground storage tanks.

Langley Research Center
National Aeronautics and Space Administration
Hampton, VA 23665
July 28, 1983

APPENDIX

COMPUTER PROGRAMS FOR CALCULATING NUCLEAR RADIATION TRANSMISSION

Symbols Used in Appendix

c	velocity of light
D	internal diameter of the steel reservoir cylinder
d	distance between the source and the detector
ds	path length between sampling planes
E_{dis}	input cutoff energy
E_f	energy of the scattered photon
E_i	initial energy of the photon prior to scattering
E_o	incident photon energy
m_o	rest mass of the electron
n_e	product of the number of test atoms per cubic centimeter and the atomic number of the test atoms
P_c	cumulative normalized probability
P_T	total transmission probability
P_t	transmission probability per photon
R_T	detector radius
r_o	classical radius of the electron
x_f, y_f, z_f	x-, y-, z-coordinates following scattering
x', y', z'	final coordinates in the photon coordinate system
α	$= E_o/m_o c^2$
θ	scattering angle
θ_f	spherical coordinate defining gamma ray direction after scattering
θ_i	spherical coordinate defining gamma ray direction before scattering
θ_s	scattering angle in the photon coordinate system
$\sigma_c(T)$	total scattering cross section

ϕ_f spherical coordinate defining gamma ray direction after scattering

ϕ_i spherical coordinate defining gamma ray direction before scattering

ϕ_s scattering angle in the photon coordinate system

Ω solid angle

Program for Calculating Energy and Transmission

Program Description

The computer program RTRACK is written in FORTRAN IV language for the Control Data CYBER 170 series digital computer system with network operating system (NOS) 1.4. The program requires 31 600 octal locations of core storage. A typical case in which 5000 trajectories are computed requires approximately 350 CPU seconds on CYBER 173.

The program models the stochastic process in which a gamma ray traverses a medium subject to multiple Compton scattering. When the detector location and size, initial gamma ray energy, dimensions and electron density of the intervening medium, and desired number of trajectories and path length for sampling are specified, the program uses a random number generator and an input probability distribution table to track the progress of each gamma ray. A separate program, described subsequently in this appendix, was developed to generate this table giving probability versus scattering angle (θ).

Once the medium is defined, the z-axis is defined from the point source to the center of the detector with the origin at the source. Equally spaced planes are constructed perpendicular to this axis between the source and the detector with the spacing defined by the input sampling frequency parameter. The initial direction of the gamma ray is determined by a random value of ϕ_i between 0 and 2π, and a random value of θ_i subject to collimation restrictions (where ϕ_i and θ_i are standard spherical coordinates). An example of such a restriction is to limit the initial gamma ray to the forward hemisphere or to the cone defined by the source and detector. A collimator with a diameter of 7/16 in. and a length of 1 27/32 in. was used.

The gamma ray is allowed to travel along the initial trajectory until this trajectory intersects the first sampling plane. The transmission probability is computed using equation (A1)

$$P_t = 1 - \sigma_c(T)n_e \, ds \tag{A1}$$

where n_e (given by the product of the number of test atoms per cubic centimeter and the atomic number of the test atoms, i.e., $n_e = nZ$) is the electron density for the appropriate material (air, water, or the steel wall), ds is the path length

between sampling planes, and $\sigma_c(T)$ is the total scattering cross section defined by equation (A2)

$$\sigma_c(T) = \int_0^{2\pi} d\phi \int_0^{\pi} \frac{d\sigma}{d\Omega} \sin\theta \, d\theta \tag{A2}$$

This transmission probability is compared with a uniformly distributed random number between 0 and 1 to determine if a collision occurs. If the random number is smaller, no collision occurs, and the gamma ray continues along the same path until it intersects the next sampling plane. Otherwise, a collision occurs, and a new path is determined.

The scattering direction is determined by assigning a random value between 0 and 2π to ϕ_s and by defining θ_s with a uniformly distributed random number and linear interpolation within the input probability distribution table. The z'-axis is defined as the direction of motion prior to the collision, and the new direction in the primed coordinate system is computed using the rotation matrices in equation (A3).

$$\begin{Bmatrix} x' \\ y' \\ z' \end{Bmatrix} = \begin{bmatrix} \cos\phi_s & \sin\phi_s & 0 \\ -\sin\phi_s & \cos\phi_s & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\theta_s & \sin\theta_s \\ 0 & -\sin\theta_s & \cos\theta_s \end{bmatrix} \begin{Bmatrix} 0 \\ 0 \\ 1 \end{Bmatrix} \tag{A3}$$

These coordinates are then transformed back to the laboratory coordinate system using the rotation matrices in equation (A4).

$$\begin{Bmatrix} x_f \\ y_f \\ z_f \end{Bmatrix} = \begin{bmatrix} \cos\phi_i & \sin\phi_i & 0 \\ -\sin\phi_i & \cos\phi_i & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\theta_i & \sin\theta_i \\ 0 & -\sin\theta_i & \cos\theta_i \end{bmatrix} \begin{Bmatrix} x' \\ y' \\ z' \end{Bmatrix} \tag{A4}$$

where θ_i and ϕ_i define the gamma ray direction before scattering. The final direction in the laboratory coordinate system following the collision is then given by equations (A5)

$$\left.\begin{aligned} \phi_f &= \tan^{-1}\left(\frac{x_f}{y_f}\right) \\ \theta_f &= \tan^{-1}\left(\frac{\sqrt{x_f^2 + y_f^2}}{z_f}\right) \end{aligned}\right\} \tag{A5}$$

The gamma ray then travels along this trajectory until it intersects the next sampling plane. Following the scattering, the energy is reduced according to equation (A6)

$$E_f = \frac{E_i}{1 + \alpha(1 - \cos\theta_s)} \tag{A6}$$

where E_i is the initial energy and $\alpha = \dfrac{E_o}{m_o c^2} = \dfrac{E_o}{511}$

This process is repeated for each sampling plane, and the gamma ray is said to reach the detector if, for the final plane, $x_f^2 + y_f^2 \leq R_T^2$, where R_T is the detector radius. The gamma ray is dropped from further consideration if, for any sampling plane, one of the following inequalities is satisfied: $x_f \geq \frac{1}{2} d$, $y_f \geq \frac{1}{2} d$, $z_f \leq 0$, or $E_f \leq E_{dis}$, where d is the distance between the source and the detector, and E_{dis} is an input cutoff energy.

After a gamma ray has reached the detector or has been dropped from further consideration, a new trajectory is initiated. This process is repeated until the desired number of trajectories has been completed. The total transmission probability is then defined by equation (A7)

$$P_T = \frac{\text{Gamma rays reaching the detector}}{\text{Total number of trajectories}} \tag{A7}$$

Description of FORTRAN Variables

The following list contains a description of the significant FORTRAN variables appearing in the program. The dimension of each array is in parentheses beside the variable. Each variable is also identified as I, input variable; P, program variable; or O, output variable.

APPENDIX

FORTRAN Variable	Type	Description
DIAM	P	Diameter of detector, cm
DS	I	Spacing between sampling planes, cm
EFINAL	I	Cutoff energy, keV
ENER	P	Instantaneous gamma ray energy, keV
EZERO	I	Initial energy, keV
NHIT	O	Number of gamma rays counted by the detector
NNE	O	Number of gamma rays with energy less than cutoff value
NNX	O	Number of gamma rays exiting in x-direction
NNY	O	Number of gamma rays exiting in y-direction
NNZ	O	Number of gamma rays exiting in z-direction
NTRY	I	Total number of trajectories
NZ	I	Electron density of the interior medium, electrons/cm^3
PHI	P	ϕ_i of path before scattering
PHIF	P	ϕ_f of path after scattering
PHIS	P	ϕ_s, scattering angle
PTM	P	Transmission probability for the interior medium
PTS	P	Transmission probability for the steel wall
R	P	Random number uniformly distributed between 0 and 1
SEP	I	Distance between source and detector, in.
TDIAM	I	Diameter of detector, in.
THETA	P	θ_i of path before scattering
THETAF	P	θ_f of path after scattering
THETAL	P	Limiting value of θ_i due to collimation
THETAS	P	θ_s, scattering angle
THIK	I	Thickness of steel wall, in.
TRANS	O	Total transmission probability

TSCAT	P	Total scattering cross section
XFINAL	P	z-coordinate of the detector
XWALL1	P	z-coordinate of first steel/medium interface
XWALL2	P	z-coordinate of second steel/medium interface
XX(180)	I	Probability distribution table, θ values
YY(180)	I	Probabililty distribution table, P_c values

Program Listings

Program RTRACK.- RTRACK, the main program, performs all input and output operations, controls the flow through the sampling planes, and calls the routines which perform the scattering transformations. A listing of RTRACK follows.

```
      PROGRAM RTRACK(OUTPUT,INPUT,TAPE6=OUTPUT,TAPE5=INPUT,TAPE1,
     1 TAPE2)
      COMMON PI,RO,ALPHA
      REAL NZDX,NZ
      DIMENSION XX(180),YY(180)
      DIMENSION ENN(5),CON(5),PHN(5)
      EXTERNAL FUN
      DATA ENN/200.,300.,400.,500.,600./
      DATA CON/10.57,9.19,8.23,7.52,6.96/
      DATA PHN/2.23,0.66,0.29,0.16,0.10/
    1 FORMAT(1H1,10X,5HRZERO,10X,5HALPHA,9X,6HWALL 1,9X,6HWALL 2,
     1 7X,8HDISTANCE,
     1 4X,11HTARGET DIAM,2X,27HTOT SCAT CROSS SECTION (PT)///)
    2 FORMAT(1H1,5X,16HELECTRON DENSITY,5X,15H  PATH LENGTH  ,
     1 15X,1HP///)
    3 FORMAT(1H0,1PE15.4,1PE15.4,1PE15.4,1PE15.4,1PE15.4,
     1 1PE15.4,1PE29.4)
    4 FORMAT(//1H0,10X,*NZ = *,1PE16.4)
    5 FORMAT(//1H0,14X,2HDS,8X,6HN Z DS)
    6 FORMAT(1H0,1PE16.4,1PE14.4)
    7 FORMAT(*0 HITS,MISSES,X-OUT,Y-OUT,E-OUT,Z-OUT*,6I5)
    8 FORMAT(1H0,1PE21.4,1PE20.4,1PE16.4)
      PI = ACOS(-1.)
      THETAS = 0.
      PHIS = 0.
      THETAF = 0.
      PHIF = 0.
C
C         READ PROBABILITY TABLE AND GENERATE NORMALIZED
C         CUMULATIVE DISTRIBUTION
C
      DO 10 I=1,180
      READ(2) XX(I),YY(I)
   10 CONTINUE
      SUM = 0.
      DO 20 I=1,180
      SUM = SUM + YY(I)
      YY(I) = SUM
      XX(I) = XX(I) + PI/360.
   20 CONTINUE
      DO 30 I=1,180
      YY(I) = YY(I)/SUM
```

```
   30 CONTINUE
      RO = 2.818E-13
   40 READ(5,*) EZERO,EFINAL
      IF(EOF(5).NE.0) GO TO 999
      ALPHA = EZERO/511.
      READ(5,*) THIK,SEP
      XINIT = 0.
      XWALL1 = 2.54*THIK
      XWALL1 = 2.*XWALL1
      XWALL2 = 2.54*(SEP - THIK)
      XWALL2 = 2.*XWALL2
      XFINAL = 2.54*SEP
      XFINAL = 2.*XFINAL
      READ(5,*) TDIAM
      DIAM = 2.54*TDIAM
      TSCAT = GLEG15(0.,PI,FUN)
      WRITE(6,1)
      WRITE(1,2)
      WRITE(6,3) RO,ALPHA,XWALL1,XWALL2,XFINAL,DIAM,TSCAT
      READ(5,*) NZ,NTRY
      WRITE(6,6) NZ
      READ(5,*) DS
      NZDX = NZ*DS
      WRITE(6,5)
      WRITE(6,6) DS,NZDX
      PTM = EXP(-TSCAT*NZDX)
      STELNZ = 22.07E+23
      PTS = EXP(-TSCAT*STELNZ*DS)
      NMISS = 0
      NHIT = 0
      NNX = 0
      NNY = 0
      NNE = 0
      NNZ = 0
      THETAL = PI/2.
      THETAL = ATAN(7./59.)
      NSCAT = 0
C
C         BEGIN MAIN LOOP OVER TRIALS
C
      DO 150 I=1,NTRY
      ENER = EZERO
      ZA = XINIT
      XA = 0.
      YA = 0.
      THETAI = RANF(T)*THETAL
      PHII = RANF(T)*2.*PI
      THETA = THETAI
      PHI = PHII
   50 CONTINUE
      R = RANF(T)
      PT = PTM
      IF(ZA.LT.XWALL1) PT = PTS
      IF(ZA.GT.XWALL2.AND.ZA.LT.XFINAL) PT = PTS
C
C         TEST TO SEE IF A SCATTER OCCURS
C
      IF(R.LT.PT) GO TO 100
C
C         YES, THERE IS A SCATTER
C
      NSCAT = NSCAT + 1
      DO 60 K=1,4
      IF(ENER.GE.ENN(K).AND.ENER.LE.ENN(K+1)) GO TO 70
```

```
   60 CONTINUE
      K = 4
   70 CONTINUE
      CONX = CON(K) + (CON(K+1) - CON(K))*(ENER - ENN(K))
     1 /(ENN(K+1) - ENN(K))
      PHNX = PHN(K) + (PHN(K+1) - PHN(K))*(ENER - ENN(K))
     1 /(ENN(K+1) - ENN(K))
      FRACN = PHNX/(PHNX + CONX)
      FRACN = 1. - FRACN
      R = 0.
      IF(ZA.LT.XWALL1) R = RANF(T)
      IF(ZA.GT.XWALL2) R = RANF(T)
C
C         TEST FOR PHOTOELECTRIC EFFECT
C
      IF(FRACN.LT.R) GO TO 130
C
C         COMPTON SCATTER
C
      R = RANF(T)
      DO 80 K=1,179
      IF(YY(K).LE.R.AND.YY(K+1).GE.R) GO TO 90
   80 CONTINUE
      K = 179
   90 CONTINUE
      THETAS = XX(K) + (XX(K+1) - XX(K))*(R - YY(K))/
     1 (YY(K+1) - YY(K))
      PHIS = RANF(T)*2.*PI
      THETA = THETA*180./PI
C
C         CONVERT ALL ANGLES TO DEGREES
C
      PHI = PHI*180./PI
      THETAS = THETAS*180./PI
      PHIS = PHIS*180./PI
C
C         GET NEW DIRECTION
C
      CALL POST(THETA,PHI,THETAS,PHIS,THETAF,PHIF)
C
C         CHANGE ANGLES BACK TO RADIANS
C
      THETA = THETAF*PI/180.
      PHI = PHIF*PI/180.
      THETAS = THETAS*PI/180.
C
C         DECREMENT ENERGY
C
      ENER = ENER/(1. + ALPHA*(1. - COS(THETAS)))
  100 CONTINUE
C
C        INCREMENT PATH
C
      ZA = ZA + DS*COS(THETA)
      XA = XA + DS*SIN(THETA)*COS(PHI)
      YA = YA + DS*SIN(THETA)*SIN(PHI)
C
C         TEST FOR OUTSIDE X-BOUNDS
C
      IF(ABS(XA).GT.XFINAL/2.) GO TO 110
C
C         TEST FOR OUTSIDE Y-BOUNDS
```

```
C
170          IF(ABS(YA).GT.XFINAL/2.) GO TO 120
C
C              TEST FOR ENERGY LESS THAN CUTOFF
C
             IF(ENER.LT.EFINAL) GO TO 130
175          DZ = ABS(XFINAL - ZA)
C
C              TEST FOR OUTSIDE Z-BOUNDS
C
             IF(ZA.LT.0.) GO TO 140
180       C
C              TEST FOR PROXIMITY TO DETECTOR
C
             IF(DZ.GT.4.*DS) GO TO 50
             DHIT = SQRT(YA**2 + XA**2)
185       C
C              TEST FOR A HIT
C
             IF(DHIT.LT.(DIAM/2.)) NHIT = NHIT + 1
             IF(DHIT.GE.(DIAM/2.)) NMISS = NMISS + 1
190          GO TO 150
           110 CONTINUE
             NNX = NNX + 1
             NMISS = NMISS + 1
             GO TO 150
195        120 CONTINUE
             NNY = NNY + 1
             NMISS = NMISS + 1
             GO TO 150
           130 CONTINUE
200          NNE = NNE + 1
             NMISS = NMISS + 1
             GO TO 150
           140 CONTINUE
             NNZ = NNZ + 1
205          NMISS = NMISS + 1
           150 CONTINUE
             WRITE(6,7) NHIT,NMISS,NNX,NNY,NNE,NNZ
             TRANS = FLOAT(NTRY - NMISS)/FLOAT(NTRY)
             WRITE(1,8) NZ,DS,TRANS
210          GO TO 40
999  CONTINUE
     STOP
     END
```

Function GLEG15.- Function GLEG15 determines the value of a definite integral using a 15-point Gauss-Legendre quadrature scheme. It is used in determining the value of TSCAT.

```
      FUNCTION GLEG15(A,B,FUNCT)
C
C FUNCTION GLEG15 EVALUATES INTEGRALS OF THE FORM
C
C          B
C     INTEGRAL [G(X)] DX
C          A
C
C WHERE A AND B ARE FINITE USING 15 POINT GAUSS-LEGENDRE QUADRATURE.
C
C USE
C
C     VALUE=GLEG15(A,B,FUNCT)
C
C           A......THE LOWER LIMIT OF INTEGRATION. MUST BE FINITE.
C
C           B......THE UPPER LIMIT OF INTEGRATION. MUST BE FINITE.
C
C           FUNCT..THE NAME OF AN EXTERNAL SUBROUTINE TO CALCULATE THE
C                  VALUE OF G(X) AT SPECIFIED POINTS.
C                  SUBROUTINE FUNCT(F,N)
C                             F..ON INPUT THE N VALUES OF X
C                                ON OUTPUT THE N VALUES G(X)
C                                THAT IS, F(I)=G[F(I)] I=1,...,N
C                             N..THE NUMBER OF FUNCTION EVALUATIONS.
C                  THE NAME GIVEN FUNCT MUST APPEAR IN AN EXTERNAL
C                  STATEMENT IN THE CALLING PROGRAM
      DIMENSION W(8),R(8),F(15)
      DATA R/0.                ,.201194093997434,.394151347077563,
     *       .570972172608538,.724417731360170,.848206583410427,
     *       .937273392400705,.987992518020485/
      DATA W/.202578241925561,.198431485327111,.186161000015562,
     *       .166269205816993,.139570677926154,.107159220467171,
     *       .703660474881081E-01,.307532419961172E-01/
      C=(B-A)*.5
      D=(B+A)*.5
      F(1)=D
      DO 10 I=2,8
      S=R(I)*C
      F(I)=D+S
   10 F(I+7)=D-S
      CALL FUNCT(F,15)
      D=W(1)*F(1)
      DO 20 I=2,8
   20 D=D+W(I)*(F(I)+F(I+7))
      GLEG15=C*D
      RETURN
      END
```

Subroutine FUN.- Subroutine FUN is required by GLEG15 to compute the integrand of the definite integral, in this case the Klein-Nishina formula.

```
      SUBROUTINE FUN(F,N)
      DIMENSION F(1)
      COMMON PI,RO,ALPHA
      DO 10 I=1,N
      THETA = F(I)
      ANS = 2.*PI*RO**2*SIN(THETA)
      TMP1 = 1. + ALPHA*(1. - COS(THETA))
      TMP2 = 1. + COS(THETA)**2
      TMP3 = (1. - COS(THETA))**2
      ANS = ANS*(1./TMP1)**2*(TMP2/2.)
     1 *(1. + ALPHA**2*TMP3/(TMP2*TMP1))
      F(I) = ANS
   10 CONTINUE
      RETURN
      END
```

Subroutine POST.- Subroutine POST computes the new direction of a trajectory following a scatter by defining and multiplying the appropriate rotation matrices.

```
      SUBROUTINE POST(T1,P1,T2,P2,T3,P3)
      DIMENSION A(4,4)
      CALL UROTOC(1,T2,A)
      CALL UAPPLY(0.,0.,1.,A,X,Y,Z)
      CALL UROTOC(3,P2,A)
      CALL UAPPLY(X,Y,Z,A,X,Y,Z)
      CALL UROTOC(1,T1,A)
      CALL UAPPLY(X,Y,Z,A,X,Y,Z)
      CALL UROTOC(3,P1,A)
      CALL UAPPLY(X,Y,Z,A,X,Y,Z)
      P3=ATAN(X/Y)*180./3.14159265
      IF (X.LT.0..AND.Y.LT.0.) P3=360.-P3
      IF (X.LT.0..AND.Y.GE.0.) P3=180.-P3
      IF (X.GE.0..AND.Y.GE.0.) P3=90.+P3
      IF (X.GE.0..AND.Y.LT.0.) P3=-P3
      R=SQRT(X*X+Y*Y)
      IF (Y.LT.0.) R=-R
      T3=ATAN(R/Z)*180./3.14159265
      IF (R.LT.0..AND.Z.LT.0.) T3=180.-T3
      IF (R.LT.0..AND.Z.GT.0.) T3=-T3
      IF (R.GE.0..AND.Z.GE.0.) T3=T3
      IF (R.GE.0..AND.Z.LT.0.) T3=180.+T3
      RETURN
      END
```

Subroutines UAPPLY, UROTOC, UCLR, and UIDENT.- Subroutines UAPPLY, UROTOC, UCLR, and UIDENT are utility routines required by subroutine POST.

```
      SUBROUTINE UAPPLY (X,Y,Z, A, U,V,W)
C
C POST-MULTIPLIES A POINT VECTOR (X Y Z) BY 4*4
C TRANSFORMATION MATRIX A TO GIVE (U V W).
C
      REAL A(4,4)
C H IS THE HOMOGENEOUS CO-ORDINATE, WHICH IS USED
C TO EFFECT VARIOUS PROJECTIONS.  IN THIS ROUTINE,
C MATRIX A IS ASSUMED TO PRODUCE ONLY AFFINE
C TRANSFORMATIONS, WHICH MEANS THE LAST COLUMN OF
C MATRIX A IS (0 0 0 1), SO H IS UNITY.  BUT IT'S
C HERE IN CASE ONE WISHES TO CHANGE IT...
C (UNCOMMENT NEXT LINE AND DIVIDE X, Y, AND Z BY H.)
C     H = X*A(1,4) + Y*A(2,4) + Z*A(3,4) + A(4,4)
      U = (X*A(1,1) + Y*A(2,1) + Z*A(3,1) + A(4,1))
      V = (X*A(1,2) + Y*A(2,2) + Z*A(3,2) + A(4,2))
      W = (X*A(1,3) + Y*A(2,3) + Z*A(3,3) + A(4,3))
      RETURN
      END
```

```
      SUBROUTINE UROTOC (IAXIS, ANGLE, A)
C
C CONSTRUCTS A 4*4 MATRIX THAT PERFORMS A ROTATION
C ABOUT X-, Y-, OR Z-AXIS OF ANY ANGLE.
C
C INPUTS --
C   IAXIS     1=X  2=Y  3=Z
C   ANGLE     AMOUNT OF ROTATION, IN CURRENT UNITS.  MEASURED
C             POSITIVE BY RIGHT-HAND RULE.
C
      REAL A(4,4)
      COMMON /TORADS/ ANGFAC
      DATA ANGFAC /.01745329252/
C
      IF (IAXIS .LE. 0 .OR. IAXIS .GE. 4)  GO TO 40
      CALL UCLR (A)
      A(4,4) = 1.
      RAD = ANGFAC * ANGLE
      COSANG = COS (RAD)
      SINANG = SIN (RAD)
      GO TO (10, 20, 30), IAXIS
   10 A(1,1) = 1.
      A(2,2) = COSANG
      A(3,2) = - SINANG
      A(2,3) = SINANG
      A(3,3) = COSANG
      RETURN
   20 A(2,2) = 1.
      A(1,1) = COSANG
```

```
      A(3,1) = SINANG
      A(1,3) = - SINANG
      A(3,3) = COSANG
      RETURN
   30 A(3,3) = 1.
      A(1,1) = COSANG
      A(2,1) = - SINANG
      A(1,2) = SINANG
      A(2,2) = COSANG
      RETURN
   40 CALL UIDENT (A)
      RETURN
      END
```

```
      SUBROUTINE UCLR (A)
C
C CLEARS 4*4 MATRIX A TO ZEROS.
C
      REAL A(4,4)
      DO 10 I=1,4
        DO 10 J=1,4
   10     A(I,J) = 0.
      RETURN
      END
```

```
      SUBROUTINE UIDENT (A)
C
C MAKES 4*4 IDENTITY MATRIX A.
C
      REAL A(4,4)
      CALL UCLR (A)
      DO 10 I=1,4
   10   A(I,I) = 1.
      RETURN
      END
```

APPENDIX

Program Usage

All program input is accomplished using FORTRAN list directed reads. The data may appear anywhere in the field, and when more than one item is specified, the data items are separated by commas. The first items in a data set are the initial gamma ray energy and the cutoff energy in keV. If, as a result of scattering, the gamma ray energy drops below the value of the cutoff energy, it is not counted by the detector. The next data items are the thickness of the container walls and the total distance between the source and the detector, in inches. These data are followed by the diameter of the detector in inches. These data items are illustrated in the sketch below:



The next data items include the electron density of the interior region (electrons/ cm^3) and the total number of gamma rays to be included in the run. The final data item is the path length to be used between sampling planes, in centimeters. If additional cases are desired in a single program execution, the entire data set must be repeated for each case. Sample data for a single program execution to analyze two cases where the interior region contains air and water are illustrated below:

```
662.0, 200.0                (EZERO, EFINAL)
0.566, 8.0                  (THIK, SEP)
2.0                         (TDIAM)
3.818E + 20, 5000           (NZ, NTRY)
0.25                        (DS)
662.0, 200.0                (Repeat for second case)
0.566, 8.0
2.0
3.346E + 23, 5000
0.25
```

In addition to the data described above, the program also requires a table containing probability versus scattering angle (θ). These data should exist as an unformatted file on unit TAPE2.

APPENDIX

The program generates formatted output on two files. File TAPE6 contains a record of the program input as well as the number of gamma rays counted by the detector. For those gamma rays not counted, TAPE6 provides information as to whether they were lost because they exceeded the limiting dimensions in the x-, y-, or z-directions or they fell below the cutoff energy. File TAPE1 provides a brief summary of total transmission probabilities for each case considered.

Program for Generating Scattering Angle Distribution

Program TDIST2 was written to generate the probability distribution table required by program RTRACK. A listing of program TDIST2 is included in this section of the appendix. Note that function GLEG15 and subroutine FUN, described with program RTRACK, are also required by this program.

Program TDIST2 calculates the probability spectrum given by the following expression at intervals of 1°:

$$\frac{d\sigma}{d\Omega} = r_o^2\left[\frac{1}{1 + \alpha(1 - \cos\theta)}\right]^2\left(\frac{1 + \cos^2\theta}{2}\right)\left\{1 + \frac{\alpha^2(1 - \cos\theta)^2}{(1 + \cos^2\theta)\left[1 + \alpha(1 - \cos\theta)\right]}\right\} \qquad \text{(A8)}$$

where $r_o = 2.818 \times 10^{-13}$ cm, and $\alpha = E_o/m_oc^2$. This equation follows from equation (8). Output from the program is written as an unformatted file to TAPE2 and is displayed in figure (A1). The normalized cumulative probability distribution versus θ corresponding to this spectrum is shown in figure (A2). Figure (A3) illustrates the results of using the distribution in figure (A2) to generate the frequency spectrum for 10 000 scattering events. As shown, both figures (A1) and (A3) are normalized to 10 000 events.

```
      PROGRAM TDIST2(OUTPUT,INPUT,TAPE6=OUTPUT,TAPE5=INPUT,
     1 TAPE2)
      COMMON PI,RO,ALPHA
      EXTERNAL FUN
      PI = ACOS(-1.)
      RO = 2.818E-13
      ALPHA = 662./511.
      TSCAT = GLEG15(0.,PI,FUN)
      WRITE(6,1)
    1 FORMAT(1H1,10X,5HRZERO,10X,5HALPHA,
     1 2X,27HTOT SCAT CROSS SECTION (PT)///)
      WRITE(6,2) RO,ALPHA,TSCAT
    2 FORMAT(1H0,1PE15.4,1PE22.4,1PE29.4)
      NUM = 180.
      DO 100 I=1,NUM
      THETA = PI*FLOAT(I-1)/180.
      THETAP = THETA + PI/180.
      PSCAT = GLEG15(THETA,THETAP,FUN)
      WRITE(2) THETA,PSCAT
      WRITE(6,3) THETA,PSCAT
    3 FORMAT(2E16.8)
  100 CONTINUE
      CONTINUE
      STOP
      END
```



Figure A1.- Probability distribution from program TDIST2.



Figure A2.- Normalized cumulative probability distribution as a function of θ.



Figure A3.- Angular distribution of the scattered photons.

REFERENCES

1. Evans, Robley D.: The Atomic Nucleus. McGraw-Hill Book Co., Inc., c.1955.

2. Davisson, Charlotte Meaker; and Evans, Robley D.: Gamma-Ray Absorption Coefficients. Rev. Mod. Phys., vol. 24, no. 2, Apr. 1952, pp. 79-107.

3. Grodstein, Gladys White: X-ray Attenuation Coefficients From 10 kev to 100 Mev. NBS Circ. 583, U.S. Dep. Com., Apr. 30, 1957.

4. Klein, O.; and Nishina, Y.: Uber die Streuung von Strahlung durch freie Elektronen nach der neuen relativistischen Quantendynamik von Dirac. Z. Phys., Bd. 52, 1929, pp. 853-868.

5. Heitler, W.: The Quantum Theory of Radiation, Third ed. Clarendon Press (Oxford), 1954.

6. Goldstein, Herbert: Fundamental Aspects of Reactor Shielding. Addison-Wesley Pub. Co., Inc., c.1959.

7. Alder, Berni; Fernbach, Sidney; and Rotenberg, Manuel, eds.: Methods in Computational Physics. Volume 1. Academic Press, Inc., 1963.

TABLE I.- SUMMARY OF EXPERIMENTAL VALUES FOR $I_t(air)/I_t(water)$ FOR FULL-SCALE MODEL

E_{dis}, keV	Air pressure, atm	$\frac{I_t(air)}{I_t(water)}$
200	1	15.60 ± 0.60
511	1	17.24 ± 1.40

TABLE II.- SUMMARY OF CALCULATED VALUES[a] FOR $I_t(air)/I_t(water)$ UNDER VARIOUS OPERATING CONDITIONS FOR FULL-SCALE MODEL

(a) "Ideal" narrow beam geometry (no multiple scattering)

E_{dis}, keV	Air pressure, atm	$\frac{I_t(air)}{I_t(water)}$
662	1	20.17
662	408.2	5.26

(b) "Poor" broad beam geometry (multiple scattering allowed)

E_{dis}, keV	Air pressure, atm	$\frac{I_t(air)}{I_t(water)}$
200	1	15.98 ± 1.50
200	408.2	4.75 ± 0.51
511	1	17.68 ± 1.95
511	408.2	5.01 ± 0.64

[a]These values were calculated for the following geometrical conditions:

Collimator hole diameter = 0.438 in.
Collimator depth = 1.844 in.
Outer diameter of steel vessel = 16.000 in.
Inner diameter of steel vessel = 13.736 in.
NaI crystal dimensions = 2-in. diameter × 2-in. height

TABLE III.- COMPARISON BETWEEN EXPERIMENTAL RESULTS AND CALCULATED VALUES FOR $I_t(air)/I_t(water)$ FOR HALF-SCALE MODEL

E_{dis}, keV	Air pressure, atm	$\frac{I_t(air)}{I_t(water)}$	
		Experimental value	Calculated value
200	1	3.60 ± 0.02	3.38 ± 0.23
511	1	3.99 ± 0.02	3.84 ± 0.28



L-83-1290

Figure 1.- Photograph of cylinder assembly for storage of cooling water.



Figure 2.- Relative importance of three major types of gamma ray interactions as a function of photon energy and absorber atomic number (ref. 1).



Figure 3.- Geometrical details of the gamma ray transmission computation problem. Linear dimensions in inches.



Figure 4.- Compton scattering effect. Paths of incident and scattered photons define scattering plane. Path of recoiling electron also lies in same plane.



Figure 5.- Schematic diagram of experimental system for monitoring presence or absence of water in test cylinder.



Figure 6.- Variation in counting rate as water is discharged from full-scale prototype water cylinder. (E_{dis} = 511 keV.)



Figure 7.- Electronic circuit for the water level alarm system.

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TM-85651		

4. Title and Subtitle	5. Report Date
DEVELOPMENT OF A NUCLEAR TECHNIQUE FOR MONITORING WATER LEVELS IN PRESSURIZED VESSELS	September 1983
	6. Performing Organization Code: 505-33-53-14
7. Author(s): Jag J. Singh, William T. Davis, and Gerald H. Mall	8. Performing Organization Report No.: L-15642
	10. Work Unit No.
9. Performing Organization Name and Address: NASA Langley Research Center, Hampton, VA 23665	11. Contract or Grant No.
	13. Type of Report and Period Covered: Technical Memorandum
12. Sponsoring Agency Name and Address: National Aeronautics and Space Administration, Washington, DC 20546	14. Sponsoring Agency Code

15. Supplementary Notes

Jag J. Singh and William T. Davis: Langley Research Center, Hampton, Virginia.
Gerald H. Mall: Computer Sciences Corporation, Hampton, Virginia.

16. Abstract

A new technique for monitoring water levels in pressurized stainless steel cylinders has been developed. It is based on differences in attenuation coefficients of water and air for Cs^{137} (662 keV) gamma rays. Experimentally observed gamma ray counting rates with and without water in model reservoir cylinders have been compared with corresponding calculated values for two different gamma ray detection threshold energies. Calculated values include the effects of multiple scattering and attendant gamma ray energy reductions. The agreement between the measured and calculated values is reasonably good. Computer programs for calculating angular and spectral distributions of scattered radiation in various media are included.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Gamma rays Narrow beam geometry Broad beam geometry Attenuation coefficient Photoelectric effect Compton scattering cross section Pair production cross section Multiple scattering Radiation detectors Monte Carlo simulation Pressurized vessels	Unclassified - Unlimited Subject Category 35

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price
Unclassified	Unclassified	33	A03

For sale by the National Technical Information Service, Springfield, Virginia 22161

NASA-Langley, 1983

National Aeronautics and
Space Administration

Washington, D.C.
20546

Official Business
Penalty for Private Use, $300

THIRD-CLASS BULK RATE

Postage and Fees Paid
National Aeronautics and
Space Administration
NASA-451

NASA

POSTMASTER: If Undeliverable (Section 158
Postal Manual) Do Not Return